


Parallel Prefix Sum on the GPU (Scan)

Presented by Adam O'Donovan

Slides adapted from the online course slides for
ME964 at Wisconsin taught by Prof. Dan Negrut
and from slides Presented by David Luebke



Parallel Prefix Sum (Scan)



- Definition:

The all-prefix-sums operation takes a binary associative operator \oplus with identity I , and an array of n elements

$$[a_0, a_1, \dots, a_{n-1}]$$

and returns the ordered set

$$[I, a_0, (a_0 \oplus a_1), \dots, (a_0 \oplus a_1 \oplus \dots \oplus a_{n-2})].$$

- Example:

if \oplus is addition, then scan on the set

$$[3 \ 1 \ 7 \ 0 \ 4 \ 1 \ 6 \ 3]$$

returns the set

$$[0 \ 3 \ 4 \ 11 \ 11 \ 15 \ 16 \ 22]$$

Exclusive scan: last input element is not included in the result

Applications of Scan



- Scan is a simple and useful parallel building block
 - Convert recurrences from sequential ...

```
for(j=1;j<n;j++)  
    out[j] = out[j-1] + f(j);
```
 - ... into parallel:

```
forall(j) in parallel  
    temp[j] = f(j);  
scan(out, temp);
```
- Useful in implementation of several parallel algorithms:
 - radix sort
 - quicksort
 - String comparison
 - Lexical analysis
 - Stream compaction
 - Polynomial evaluation
 - Solving recurrences
 - Tree operations
 - Histograms
 - Etc.

Scan on the CPU



```
void scan( float* scanned, float* input, int length)
{
    scanned[0] = 0;
    for(int i = 1; i < length; ++i)
    {
        scanned[i] = scanned[i-1] + input[i-1];
    }
}
```


- Just add each element to the sum of the elements before it
- Trivial, but sequential
- Exactly $n-1$ adds: optimal in terms of work efficiency

Parallel Scan Algorithm: Solution One


Hillis & Steele (1986)



- Note that a implementation of the algorithm shown in picture requires two buffers of length n (shown is the case $n=8=2^3$)
- Assumption: the number n of elements is a power of 2: $n=2^M$




The Plain English Perspective



- First iteration, I go with stride $1=2^0$
 - Start at $x[2^M]$ and apply this stride to all the array elements before $x[2^M]$ to find the mate of each of them. When looking for the mate, the stride should not land you before the beginning of the array. The sum replaces the element of higher index.
 - This means that I have 2^M-1 additions
- Second iteration, I go with stride $2=2^1$
 - Start at $x[2^M]$ and apply this stride to all the array elements before $x[2^M]$ to find the mate of each of them. When looking for the mate, the stride should not land you before the beginning of the array. The sum replaces the element of higher index.
 - This means that I have $2^M - 2^1$ additions
- Third iteration: I go with stride $4=2^2$
 - Start at $x[2^M]$ and apply this stride to all the array elements before $x[2^M]$ to find the mate of each of them. When looking for the mate, the stride should not land you before the beginning of the array. The sum replaces the element of higher index.
 - This means that I have $2^M - 2^2$ additions
- ... (and so on)

The Plain English Perspective



- Consider the k^{th} iteration (k is some arbitrary valid integer): I go with stride 2^{k-1}
 - Start at $x[2^M]$ and apply this stride to all the array elements before $x[2^M]$ to find the mate of each of them. When looking for the mate, the stride should not land you before the beginning of the array. The sum replaces the element of higher index.
 - This means that I have $2^M - 2^{k-1}$ additions
- ...
- M^{th} iteration: I go with stride 2^{M-1}
 - Start at $x[2^M]$ and apply this stride to all the array elements before $x[2^M]$ to find the mate of each of them. When looking for the mate, the stride should not land you before the beginning of the array. The sum replaces the element of higher index.
 - This means that I have $2^M - 2^{M-1}$ additions
- NOTE: There is no $(M+1)^{\text{th}}$ iteration since this would automatically put me beyond the bounds of the array (if you apply an offset of 2^M to “ $\&x[2^M]$ ” it places you right before the beginning of the array – not good...)

Hillis & Steele Parallel Scan Algorithm



- Algorithm looks like this:

```
for  $d := 0$  to  $M-1$  do
  forall  $k$  in parallel do
    if  $k - 2^d \geq 0$  then
       $x[out][k] := x[in][k] + x[in][k - 2^d]$ 
    else
       $x[out][k] := x[in][k]$ 
  endforall
  swap( $in, out$ )
endfor
```

Double-buffered version of the sum scan

Operation Count

Final Considerations




- The number of operations tally:
 - $(2^M - 2^0) + (2^M - 2^1) + \dots + (2^M - 2^k) + \dots + (2^M - 2^{M-1})$
 - Final operation count:

$$M \cdot 2^M - (2^0 + \dots + 2^{M-1}) = M \cdot 2^M - 2^M + 1 = n(\log(n) - 1) + 1$$

- This is an algorithm with $O(n \cdot \log(n))$ work
- This scan algorithm is not that work efficient
 - Sequential scan algorithm does $n-1$ adds
 - A factor of $\log(n)$ might hurt: 20x more work for 10^6 elements!
- A parallel algorithm can be slow when execution resources are saturated due to low algorithm efficiency

Hillis & Steele: Kernel Function



```
__global__ void scan(float *g_odata, float *g_idata, int n)
{
    extern __shared__ float temp[]; // allocated on invocation

    int thid = threadIdx.x;
    int pout = 0, pin = 1;

    // load input into shared memory.
    // Exclusive scan: shift right by one and set first element to 0
    temp[thid] = (thid > 0) ? g_idata[thid-1] : 0;
    __syncthreads();


    for( int offset = 1; offset < n; offset <<= 1 )
    {
        pout = 1 - pout; // swap double buffer indices
        pin = 1 - pout;

        if (thid >= offset)
            temp[pout*n+thid] += temp[pin*n+thid - offset];
        else
            temp[pout*n+thid] = temp[pin*n+thid];

        __syncthreads();
    }

    g_odata[thid] = temp[pout*n+thid]; // write output
}
```

Hillis & Steele: Kernel Function, Quick Remarks



- The kernel is very simple, which is good
- Note the nice trick that was used to swap the buffers
- The kernel only works when the entire array is processed by one block
 - One block in CUDA has 512 threads, which means I can have up to 1024 elements
 - This needs to be improved upon, can't limit solution to what's been presented so far


Improving Efficiency



- A common parallel algorithm pattern:
 - Balanced Trees*
 - Build a balanced binary tree on the input data and sweep it to and then from the root
 - Tree is not an actual data structure, but a concept to determine what each thread does at each step
- For scan:
 - Traverse down from leaves to root building partial sums at internal nodes in the tree
 - Root holds sum of all leaves (this is a reduction algorithm!)
 - Traverse back up the tree building the scan from the partial sums

Picture and Pseudocode

~ Reduction Step ~




i1 =				
	1	3	5	7
	3	7	-1	-1
	7	-1	-1	-1
i2 =				
	0	2	4	6
	1	5	-1	-1
	3	-1	-1	-1

NOTE: "-1" entries indicate no-ops

```

for k=0 to M-1
  offset = 2k
  for j=1 to 2M-k-1 in parallel do
    x[j * 2k+1 - 1] = x[j * 2k+1 - 1] + x[j * 2k+1 - 2k - 1]
  endfor
endfor
    
```

Operation Count, Reduce Phase




```
for k=0 to M-1
  offset = 2k
  for j=1 to 2M-k-1 in parallel do
    x[j·2k+1-1] = x[j·2k+1-1] + x[j·2k+1-2k-1]
  endfor
endfor
```

By inspection: $\sum_{k=0}^{M-1} 2^{M-k-1} = 2^M - 1 = n - 1$

Looks promising...

The Down-Sweep Phase



NOTE: This is just a mirror image of the reduction stage. Easy to come up with the indexing scheme...

```

for k=M-1 to 0
  offset = 2k
  for j=1 to 2M-k-1 in parallel do
    dummy = x[j·2k+1-2k-1]
    x[j·2k+1-2k-1] = x[j·2k+1-1]
    x[j·2k+1-1] = x[j·2k+1-1] + dummy
  endfor
endfor
    
```

Down-Sweep Phase, Remarks



- Number of operations for the down-sweep phase:
 - Additions: $n-1$
 - Swaps: $n-1$ (each swap shadows an addition)
- Total number of operations associated with this algorithm
 - Additions: $2n-2$
 - Swaps: $n-1$
 - Looks very comparable with the work load in the sequential solution
- The algorithm is convoluted though, it won't be easy to implement
 - Kernel shown on next slide


```

01| __global__ void prescan(float *g_odata, float *g_idata, int n)
02| {
03|     extern __shared__ float temp[]; // allocated on invocation
04|
05|
06|     int thid = threadIdx.x;
07|     int offset = 1;
08|
09|     temp[2*thid] = g_idata[2*thid]; // load input into shared memory
10|     temp[2*thid+1] = g_idata[2*thid+1];
11|
12|     for (int d = n>>1; d > 0; d >>= 1) // build sum in place up the tree
13|     {
14|         __syncthreads();
15|
16|         if (thid < d)
17|         {
18|             int ai = offset*(2*thid+1)-1;
19|             int bi = offset*(2*thid+2)-1;
20|
21|             temp[bi] += temp[ai];
22|         }
23|         offset *= 2;
24|     }
25|
26|     if (thid == 0) { temp[n - 1] = 0; } // clear the last element
27|
28|     for (int d = 1; d < n; d *= 2) // traverse down tree & build scan
29|     {
30|         offset >>= 1;
31|         __syncthreads();
32|
33|         if (thid < d)
34|         {
35|             int ai = offset*(2*thid+1)-1;
36|             int bi = offset*(2*thid+2)-1;
37|
38|             float t = temp[ai];
39|             temp[ai] = temp[bi];
40|             temp[bi] += t;
41|         }
42|     }
43|
44|     __syncthreads();
45|
46|     g_odata[2*thid] = temp[2*thid]; // write results to device memory
47|     g_odata[2*thid+1] = temp[2*thid+1];
48| }

```






Bank Conflicts

- Current implementation has many ShMem bank conflicts
 - Can significantly hurt performance on current GPU hardware
 - The source of the conflicts: linear indexing with stride that is a power of 2 multiple of thread id (see below): “ $j \cdot 2^{k+1} - 1$ ”

```
for k=0 to M-1
  offset = 2k
  for j=1 to 2M-k-1 in parallel do
    x[j · 2k+1 - 1] = x[j · 2k+1 - 1] + x[j · 2k+1 - 2k - 1]
  endfor
endfor
```

- Simple modifications to current memory addressing scheme can save a lot of cycles

Bank Conflicts



- Occur when multiple threads access the same shared memory bank with different addresses
 - In our case, we have something like $2^{k+1} \cdot j - 1$
 - $k=0$: two way bank conflict
 - $k=1$: four way bank conflict
 - ...
- No penalty if all threads access different banks
 - Or if all threads access exact same address
- Recall that shared memory accesses with conflicts are serialized
 - N-bank memory conflicts lead to a set of N successive shared memory transactions

Initial Bank Conflicts on Load



- Each thread loads two shared mem data elements
- Tempting to interleave the loads (see lines 9 & 10, and 46 & 47)

```
temp[2*thid] = g_idata[2*thid];  
temp[2*thid+1] = g_idata[2*thid+1];
```


 - Thread 0 accesses banks 0 and 1
 - Thread 1 accesses banks 2 and 3
 - ...
 - Thread 8 accesses banks 16 and 17. Oops, that's 0 and 1 again...
 - Two way bank conflict, can't be easily eliminated
- Better to load one element from each half of the array

```
temp[thid] = g_idata[thid];  
temp[thid + (n/2)] = g_idata[thid + (n/2)];
```




Bank Conflicts in the tree algorithm

- When we build the sums, during the first iteration of the algorithm each thread in a half-warp reads two shared memory locations and writes one:
- Th(0,8) access bank 0



First iteration: 2 threads access each of 8 banks.


Each  corresponds to a single thread.

Like-colored arrows represent simultaneous memory accesses




Bank Conflicts in the tree algorithm

- When we build the sums, each thread reads two shared memory locations and writes one:
- Th(1,9) access bank 2, etc.



First iteration: 2 threads access each of 8 banks.


Each  corresponds to a single thread.

Like-colored arrows represent simultaneous memory accesses


Bank Conflicts in the tree algorithm



- 2nd iteration: even worse!
 - 4-way bank conflicts; for example:
Th(0,4,8,12) access bank 1, Th(1,5,9,13) access Bank 5, etc.



2nd iteration: 4 threads access each of 4 banks.

Each  corresponds to a single thread.

Like-colored arrows represent simultaneous memory accesses

Managing Bank Conflicts in the Tree Algorithm



- Use padding to prevent bank conflicts
 - Add a word of padding every 16 words.
 - Now you work with a virtual 17 bank shared memory layout
 - Within a 16-thread half-warp, all threads access different banks
 - They are aligned to a 17 word memory layout
 - It comes at a price: you have memory words that are wasted
 - Keep in mind: you should also load data from global into shared memory using the virtual memory layout of 17 banks

Use Padding to Reduce Conflicts



- After you compute a ShMem address like this:

```
Address = 2 * stride * thid;
```

- Add padding like this:


```
Address += (address >> 4); // divide by  
NUM_BANKS
```

- This removes most bank conflicts

Managing Bank Conflicts in the Tree Algorithm




Original scenario.



Modified scenario, virtual 17 bank memory layout.

Actual physical memory (true bank number)



Expanding to arbitrary sized Array



- At this point we have assumed that the array is processed by one block
- Extending to larger arrays requires splitting the array into block sized chunks
- Perform scan independantly on each chunch while storing the total sum in a new array sums.
- Another kernel then updates each block with all the sums to its left.

Expanding to arbitrary sized Array





Figure 5: Algorithm for performing a sum scan on a large array of values.

Timing results




CUDA Scan Performance



GeForce 8800 GTX, Intel Core2 Duo Extreme 2.93 GHz

GPU vs. CPU **20x**

CUDA vs. OpenGL **7x**



Scan Primitives

- Enumerate
 - Input: 2 arrays of size N composed of True False Value Pairs
 - Output: an array of size N where each element is the total number of true elements to its left.
 - Easily implemented if we represent true with 1 and false with zero
 - Simply scan the true false array

1	0	1	1	0	1
0	1	1	2	3	3



Scan Primitives

- Array Compaction
 - Input: 2 arrays of size N composed of True False Value Pairs
 - Output: An array that consists of only the values associated with a True key.
 - Simply scan the true false array
 - Everywhere the array is true we now have an index into the output array

A	B	C	D	E	F
1	0	1	1	0	1
0	1	1	2	3	3

A	C	D	F
---	---	---	---

Primitives Built Atop Scan

- Split & Split-And-Segment [great for Radix-Sort]
 - Divide input vector into two pieces, with all the elements marked *false* (0) to the left side of the output, all elements marked *true* (1) to the right side.

[1 0 1 0 0 1 0]

[0 1 0 1 1 0 1]

[0 **0** 1 **1** **2** 3 **3**]

NF = 4

[0 1 2 3 4 5 6]

[**4** 5 **5** 6 6 **6** 7]

→ Invert them


→ f = prescan (find new false indices)

→ Total number of falses


→ Thread ids

→ $t = id - f + NF$

Application: Radix Sort




- Sort 4M key-value pairs (32-bit keys): **80ms**
- Perform split operation on each bit using scan



Quad Tree Construction


- Using Scan, Radix Sort, and bit interleaving we can create a quad tree data structure
- Algorithm
 - Interleave bits of the coordinate values
 - Sort interleaved array using Radix sort
 - Find locations of boundaries of Quadtree blocks using scan
 - Store array of indexes to boundary locations

Quad Tree Construction




	x:	0	1	2	3	4	5	6	7
		000	001	010	011	100	101	110	111
y: 0	000	000000	000001	000100	000101	010000	010001	010100	010101
1	001	000010	000011	000110	000111	010010	010011	010110	010111
2	010	001000	001001	001100	001101	011000	011001	011100	011101
3	011	001010	001011	001110	001111	011010	011011	011110	011111
4	100	100000	100001	100100	100101	110000	110001	110100	110101
5	101	100010	100011	100110	100111	110010	110011	110110	110111
6	110	101000	101001	101100	101101	111000	111001	111100	111101
7	111	101010	101011	101110	101111	111010	111011	111110	111111


Results



(a)



(b)




(c)

(c) $m = 10000$, $d = 2$, VARYING K .




**Taking it one step further: completely eliminating the bank conflicts
(individual reading)**

Scan Bank Conflicts (1)




- A full binary tree with 64 leaf nodes:




- Multiple 2-and 4-way bank conflicts
- Shared memory cost for whole tree
 - 1 32-thread warp = 6 cycles per thread w/o conflicts
 - Counting 2 shared mem reads and one write ($s[a] += s[b]$)
 - $6 * (2+4+4+4+2+1) = 102$ cycles
 - 36 cycles if there were no bank conflicts ($6 * 6$)

Scan Bank Conflicts (2)



- It's much worse with bigger trees!
- A full binary tree with 128 leaf nodes
 - Only the last 6 iterations shown (root and 5 levels below)




- Cost for whole tree:
 - $12 \cdot 2 + 6 \cdot (4 + 8 + 8 + 4 + 2 + 1) = 186$ cycles
 - 48 cycles if there were no bank conflicts! $12 \cdot 1 + (6 \cdot 6)$

Scan Bank Conflicts (3)



- A full binary tree with 512 leaf nodes
 - Only the last 6 iterations shown (root and 5 levels below)



- Cost for whole tree:
 - $48 \cdot 2 + 24 \cdot 4 + 12 \cdot 8 + 6 \cdot (16 + 16 + 8 + 4 + 2 + 1) = 570$ cycles
 - 120 cycles if there were no bank conflicts!

Fixing Scan Bank Conflicts




- Insert padding every NUM_BANKS elements

```
const int LOG_NUM_BANKS = 4; // 16 banks on G80
int tid = threadIdx.x;
int s = 1;
// Traversal from leaves up to root
for (d = n>>1; d > 0; d >>= 1)
{
    if (thid <= d)
    {
        int a = s*(2*tid); int b = s*(2*tid+1)
        a += (a >> LOG_NUM_BANKS); // insert pad word
        b += (b >> LOG_NUM_BANKS); // insert pad word
        shared[a] += shared[b];
    }
}
```


Fixing Scan Bank Conflicts



- A full binary tree with 128 leaf nodes
 - Only the last 6 iterations shown (root and 5 levels below)



- No more bank conflicts!
 - Significant performance win:
 - 106 cycles vs. 186 with bank conflicts vs. 48 optimal

Fixing Scan Bank Conflicts



- It's possible to remove *all* bank conflicts
 - Just do multi-level padding
 - Example: two-level padding:

```
const int LOG_NUM_BANKS = 4; // 16 banks on G80
int tid = threadIdx.x;
int s = 1;
// Traversal from leaves up to root
for (d = n>>1; d > 0; d >>= 1)
{
    if (thid <= d)
    {
        int a = s*(2*tid); int b = s*(2*tid+1)
        int offset = (a >> LOG_NUM_BANKS); // first level
        a += offset + (offset >>LOG_NUM_BANKS); // second level
        offset = (b >> LOG_NUM_BANKS); // first level
        b += offset + (offset >>LOG_NUM_BANKS); // second level
        temp[a] += temp[b];
    }
}
```

